
Overview

You need to move a VFP application to the web.  You know in theory what you’re supposed to do, but you’ve never done this before.  What are the problems you’re likely to run into?  How do you debug them?  What security issues are you likely to face?  How will you host this application?  This session will address all these issues and more.

Caveats

In order to run the examples that come with this session as is, you’ll need a copy of Web Connection.  The demo version of Web Connection should work just fine.  If you’re using one of the other tools, you may need to make some minor changes to the code.  Please refer to the instructions with the tool of your choice to determine how to configure and run an application in that environment.
For this demo, you’ll need to define Login.htm as a document in IIS for the virtual site.  In the source code that comes with this session, the “WebContent” subdirectory is the path for the virtual site, and all files in that subdirectory belong in the virtual site.
I have not included the wc.dll file with the source code because that file is licensed.  It would go in the WC subdirectory under the WebContent subdirectory.  The application also expects a script map defined in IIS, called ISM, which points to this wc.dll file.

The path set in the Config.FPW file assumes that your copy of Web Connection is in a subdirectory of the application called WCONNECT.  Be sure to change the path in the CONFIG.FPW file to reflect your environment.

Before your application can run…

There are a number of things that can go wrong before you even get your application to run.  These are not necessarily programming problems, but problems with setting up your development environment.

You may start up your application, start up your browser and try to access the web site.  You may find that the server isn’t even seeing requests from the browser.  This could be the result of any of a number of problems:

1. You have to make sure you’ve got IIS installed on your development machine, and that you’ve setup a virtual directory for the new application, with the appropriate permissions set.  If you’re using any script maps in your application, you need to make sure those are properly configured in IIS as well. That would include your virtual directory and possibly a temporary directory, but NOT your application directory.
2. If there are any directories used by the IUSR_machine account (the Internet guest account on your machine), you need to give the appropriate permissions to the IUSR_machine account for those directories. You should NOT give IUSR_machine any access to your application directory or data directory.
3. You may need to specify a path in your application and/or your ISAPI connector, which must match.  Check that they do.

4. Depending on the mechanism you’re using for messaging, other problems may arise, such as COM configuration.  You may need to consult with others using your tool of choice to determine what your problem may be.

Once your server is seeing requests from users, you may also find that users aren’t seeing the response from the server.  This isn’t as common as the server not seeing the request, and this one is usually a programming problem of some kind.  Most often, your application isn’t sending the output you think it is, and tracing through the code will often reveal where this problem occurs.
Debugging Web Applications

The good news is, tracking down programming problems in a VFP web application doesn’t have to be any more difficult than tracking down programming problems in a LAN application.  In fact, it’s often much easier because testing your changes are often just a browser “refresh” away.  As with your LAN application, you want to set a flag that tells the application whether or not you’re running in development mode.  If you are running in development mode, any errors encountered while running your application will be handled by VFP in the same way it does with a LAN application—by displaying an error dialog and offering the choices of “Cancel”, “Ignore” or “Suspend”.  You can suspend and open the VFP debugger, and see all the same kinds of things you’re used to seeing when you debug.  You can even put a SET STEP ON anywhere in your code to bring up the debugger at any time.  Just keep in mind that you probably want to click on the “Stop” icon in your browser, so the browser doesn’t timeout while you’re looking around in the debugger. The “back” button often won’t work, depending on the browser’s configuration.

One really nice feature of web apps is that your server application doesn’t have to be running between hits from the browser.  (Just try that with a LAN app!)  You can leave the browser up on a screen you’re ready to submit, make some changes in your application, re-start the server application, and hit the submit button in the browser.

Typical Programming Errors

Aside from not sending output back to the user, here’s a list of errors a new web developer is likely to make.

Using Variables or Properties to Store Info Between Hits

Unlike the typical LAN application, where VFP is running on each workstation, in a web application, VFP is only running on the server.  In a LAN application, it’s common practice to store all kinds of settings in an application property or even global variables.  In a LAN application, those are unique to the workstation.  Because VFP is only running on the server, for a web application, this is a very bad practice.  You may set a variable or property to something for one user, and the next request could be from a different user, who then sees the value set by the previous user.  

To run the example that demonstrates this problem, first login as Barbara. Click on the “Public Variable” link and make note of the owner shown. Open a second instance of your browser and login as Tom.  Click the “Public Variable” link and note that owner in this case is “Tom”.  Go back to the first instance of your browser, and click on the “Public Variable” link again, and note that the owner that had shown, “Barbara” for this user is now showing “Tom”.

In the code example for this session, I set a global variable to the name of the user who logs in, during the Login function.  When the next user logs in, that variable is changed.  When the first user comes back and queries the contents of that variable, the value is incorrect for that user.  The Login function also creates a session variable called “Owner”, and the ShowOwner method in the code has a section commented out that shows the correct way to deal with this situation.
Using the Same Cursor Name for All Users

In a LAN application, when we want to display a subset of information, we typically use a cursor.  Cursors are unique to the workstation in a LAN application.  Once again, because we are only running VFP on the server, this is not the case in a web application.  
To demonstrate this problem with the sample code, click on the “Two users one cursor” link from the options, while logged in as “Barbara”.  Step through the invoices shown using the “Next” and “Previous” keys, and note that the owner shown for all invoices is “Barbara”.  From the instance of your browser logged in as “Tom”, click on the “Two users one cursor” link.  Step through the invoices shown and note that all the invoices are for the owner, “Tom”.  Now go back to the instance logged in as “Barbara”, and step through the invoices.  All the invoices now say the owner is “Tom”!  What happened is that after “Barbara” created the cursor for her invoices, “Tom” chose the same option and overwrote the “Barbara” cursor with invoices for “Tom”.  

This code actually demonstrates bad practice in a couple of ways.  Not only does it use one cursor that can be overwritten by other users, but it also makes assumptions about the current record pointer.  When you click on “Next” or “Previous”, it simply does a SKIP or a SKIP -1.  It would be better to send the current record number or key back to the browser as a hidden variable, then read that on the next hit and position the record pointer accordingly.

So, what do you do to avoid the problem of interfering data sets? Let’s look at some alternatives.
Requery on each Hit

The best option, if the query doesn’t take too long to run is to re-run the query for each hit.  Send this output to a cursor (the name can be the same for each hit), format the output, then close the cursor.  The advantage to this technique is there are no cleanup issues, and it’s fairly scalable.  If queries take too long to run, however, this option is not feasible.
Using Cursors

You can use cursor for temporary data, but you must use a naming scheme where users won’t interfere, or even multiple instances of the same user running different queries won’t interfere.  Something that may work in your situation is to name the cursor for the user and function.  

But remember, this is a stateless environment.  What if the user makes a request of the server and goes off to another site before the results are complete?  What if they only view some of the results.  How do you know when they’re done?  You could be left with an awful lot of open cursors hanging around, without knowing if it’s safe to close them.

One alternative to this could be to create a session variable that stores the names of all cursors that user has open.  You should periodically review open sessions, and delete any sessions that are over a certain age.  (That age will vary widely with the type of application.)  During this process, you could retrieve the list of cursors open for any session you’re about to delete, and close those cursors.

The biggest problem with this technique is that it’s the least scalable.  If you need to run more than one instance of your server, which most apps require, using cursors won’t work.  A cursor exists on a specific server, and there’s no guarantee which server will get the user’s next hit.

Using Tables

Temporary tables are a much cleaner approach than cursors, but should be used sparingly because of cleanup issues.  Again, you need a naming scheme that won’t interfere with other users or other sessions.  You could use SYS(2015) or some other naming convention.  You can safely close the table after creating the response for the user, then reopen it again for the next request.  Reopening a recently closed table is fairly quick on Windows 2000 Server and later servers because of caching.
  Your routine maintenance would then include a function that scanned through all the tables in the temporary area, deleting any beyond a certain age.

Error on the Server
In your LAN applications, you probably have some technique for handling errors that involves storing information about the error, then terminates what it was doing when the error occurred. You may try to RETURN TO MASTER, or THISFORM.RELEASE(), or maybe you just quit the application completely.  There isn’t really a “MASTER” to return to in a web application.  Nor is there a “Form” to be released.  You may want to return to a “home page”, if your application has one.  There are two things you must do in a web application when an error is encountered:

1. Send a display to the user that there was an error of some kind, optionally giving information the user can give to the web master

2. Keep the server running!

You don’t want to put up any kind of display on the server that requires human intervention.  This will effectively hang your server!  That means no WAIT WINDOWs, unless there’s a very short TIMEOUT clause, and no MESSAGEBOXes, also unless there’s a very short timeout parameter.  Chances are, no one will see these anyway, so it’s best not to use them at all.

Put up some kind of display, letting the user know there was a problem.  You can let the user navigate to another place in the application, or send the user somewhere else after a short timeout, but do let the user know what happened.

In the code sample, the method called “ServerError” shows what happens when an error goes unhandled on the server.  You can run this method from the “Error on the server” link from the demo options.  You’ll see how the error is displayed on the server, and waits for human response.
The code sample also has a section which is commented out that shows a better way to handle an error.  A message is sent back to the user, and then the user is sent back to the demo options screen after a 10 second timeout.

Status or Progress Messages on the Server

A similar mistake to the error handling situation discussed above is the display of status or progress messages on the server.  Although these don’t hang the server, the user may think the server is hung.  Using typical LAN techniques for displaying progress or status messages in a web application will only display on the server, and the user won’t have a clue about what’s happening.  In fact, if the process takes any appreciable time, the user’s browser may even timeout waiting for a response.  In addition to that, the time the server is taking to run this process, it’s unavailable to handle any other hits.
To see an example of this problem, click on the “Status message on the server” option in the example program.

One solution could be to send these requests to another server, which generates results and sends them by e-mail to the user.  This is fine in some situations, but not always.

What if the process being performed is crucial to the current online session—the user can’t leave and get their results later?  The answer to this situation is not simple.  The problem is that in a stateless system, once a response has been sent to a user, there’s no mechanism to keep the server running a process, or an automatic way for the browser to get the results back later.  

The solution is to use an asynchronous process to run the request.  For this session, I use an  example created by J. Randy Pearson for the book, “WebRAD: Building Database Applications on the Web with Visual FoxPro and Web Connection”
 which has been modified so that the application with the book is not required.
To run the example, find the “StatusMessage” method in the code and comment out the code running there now.  Remove the comment from the call to “This.SlowTasks()”, and run the “Status message on the server” option of the demo again.

How does this function work?  There are three pieces required:

1. The front end (the user running a browser)

2. The current server application
3. A second server application for processing slower requests—called the “Async Server”

The process goes like this:

· The front end makes a request of the server

· The server identifies this task as one that takes some time, and creates a record in an asynchronous request table (the “async table”), with a unique ID for the job.

· The server sends a status message back to the browser, with a timeout that triggers the browser to query the server again in a short while, sending back the job ID with the request.  (This is done in HTML with a <METAHTTP-EQUIV="Refresh" CONTENT="…”> tag that specifies the timeout period.)  Note that after this, the main server is ready to receive requests from other users.
· A server running in the background is dedicated to running these longer tasks.  It periodically checks the async table to see if any new requests have been added.  When a new record is found, it makes a note in the record that the process has been started.  Periodically, this server will update the status in the record with new information.  When the process is completed, the status is changed to reflect this, and the results are put into the job’s record.
· In the meantime, the front end sends a request about the job status to the server again.

· The server checks on the status of the job in the async table, and returns the latest progress information to the browser, with another trigger to requery the server again at a later time.

· Eventually, when the browser queries the server, the server finds the job has been completed, and send the results that are found in the async table back to the browser.

As you can see, this is one way to handle longer requests without causing the browser to timeout, and keeping the user informed of the progress.

Displaying Anything on the Server Requiring Human Intervention

I can’t stress this point enough.  Anything that displays something on the server that requires a response from a person will hang your server.  This includes the dreaded open file dialog, which can’t be trapped with error handlers.  I have a method called OpenTable that I always call before attempting to do anything with an alias on each hit.  You must use some form of this mechanism to protect yourself from the open file dialog.

There is a command called SYS(2335), which sets unattended server mode in an application.  Be careful of how you use this!  This command is only intended for situations where you are running your application as a COM server.  Turning on unattended server mode in an EXE that you run normally will cause your application to crash when it’s not started from the VFP command window!
URL Hacking

URL Hacking is when someone alters the contents of the address bar in their browser to make calls to your application in ways you didn’t intend.  This idea is probably completely foreign to you if you’ve only been dealing with LAN applications, but it’s a very real issue with web applications.  This is just one more reason why you have to carefully check all input from the user.

To run the URL Hacking example, choose the “URL Hacking” link from the demo options.  What you’ll see is a slightly different version of the Invoices screen we used in Part 1 of this session.  Here’s the code for this version of the Invoices screen:
<html>

<head>

<title>Invoices</title>

</head>

<body>

<h1 align="center">Invoices</h1>

<center>

  <p>Invoice#: <input type="text" name="txtInvNo" 

       value="<%= iif(TYPE("oInvoices")='O',oInvoices.InvNo,'0') %>"

       size="20"></p>

  <p>Owner: <input type="text" name="txtOwner" 

       value="<%= iif(TYPE("oInvoices")='O',oInvoices.OwnerName,'') %>"

       size="20"></p>

  <p>Date: <input type="text" name="txtInvDate" 

       value="<%= iif(TYPE("oInvoices")='O',oInvoices.InvDate,'  /  /    ')%>"

       size="20"></p>

  <p>Total: <input type="text" name="txtInvTotal" 

       value="<%= iif(TYPE("oInvoices")='O',oInvoices.InvTotal,'0') %>"

       size="20"></p>

  <p>

  <a href="FindInv.ism?InvNo=0">Find</a>

  &nbsp;

  <a href="DelInv.ism?InvNo=

      <%=iif(TYPE("oInvoices")='O',oInvoices.InvNo,'0') %>">Delete</a>

  &nbsp;

  <a href="SaveInv.ism?InvNo=

      <%= iif(TYPE("oInvoices")='O',oInvoices.InvNo,'0') %>">Save</a>

  &nbsp;<a href="DemoOptions.ism">Return to Options</a>

  </p>

</center>

</body>

</html>

And here’s the screen this code will show:

The difference between this form and the one we used in part 1 is that instead of buttons across the bottom, we have hyperlinks.  The hyperlinks call exactly the same functions we used for the buttons in part 1, except that with the hyperlinks, you can see what those functions are.  When I took this screen shot, I had the mouse cursor over the “Find” hyperlink.  Notice that the status bar at the bottom of the screen shows exactly where this hyperlink will take you, and with which parameters.  Note that the exact syntax used to call your code from a hyperlink may differ, depending on which web tool you are using.  Please refer to the documentation for your tool of choice for the syntax you must use.
There’s a deliberate bug in the “Find” hyperlink of this application.  When I try to enter an invoice number to find in the “Invoice#” field and click on the “Find” link, I get an error message, as shown below.

You may see this message and say to yourself, “Wait a minute!  I entered invoice number 11, not 0!  Why is it trying to find invoice 0?  But wait!  Look at the address bar!  It’s showing &InvNo=0.  What if I just change the 0 to 11 and press the Go button?”

Good question. Go ahead and try it.  Guess what?  You get the invoice screen for invoice number 11!

This URL is now also recorded in your browser’s history, so you can use the dropdown list to easily get back to the URL at any time.  This is probably not something you want your users doing, but there’s nothing you can do to prevent it.
Now, let’s try something else.  Let’s try to find invoice 2 using the same hacking technique we used to find invoice 11.  While viewing invoice 11, change the address bar so that the 11 is now a 2, and click the browser’s Go button.

We check the owner of the invoice in the Find function, and in this case, invoice 2 is for a different owner.  After 5 seconds, this message will send you back to the demo options.  From there, click on the “URL Hacking” link again.
Use the history drop down list of your browser to go back to viewing invoice 11, and click on the “Delete” link.  You’ll see the following confirmation that the invoice was deleted.

Take note of the URL in the address bar.  After 5 seconds, this message will return to the demo options.  Enter the same address in the address bar as is shown above, except change the 11 to 2, then click the “Go” button.  You’ll get the following message.

Remember that when we tried to find invoice 2, it said that invoice didn’t belong to us?  If you use the program properly, you should never be able to see invoice 2, and therefore, wouldn’t be able to delete it by clicking on the “Delete” link.  If I had programmed with that assumption in mind, I probably wouldn’t have bothered to check the owner in the delete function.  This is another case where you must check input from the user, even if you think there are restrictions on what they can do.  The only way you could possibly be trying to delete an invoice that doesn’t belong to you is if you hacked the URL.

Let’s take a closer look at the code in the DelInv function.

FUNCTION DelInv

IF not this.OpenTable("Invoices")


RETURN 

ENDIF 

lnInvNo = VAL(Request.Form('txtInvNo'))

IF EMPTY(lnInvNo)

   lnInvNo = VAL(Request.QueryString("InvNo"))

ENDIF 

LOCATE for InvNo = lnInvNo

IF OwnerName <> Session.GetSessionVar("Owner") AND NOT EOF()


this.ErrorMsg("Hacker Alert!",
       "You hacked the URL, <br><b>you bad person, you!</b>",
         ,15,"DemoOptions.ism")

ELSE 

   IF EOF()

      this.StandardPage("Locate error",
          "I couldn't find invoice "+ ;

          TRANSFORM(lnInvNo)+" and I'm not sure what to do about it",
          ,5,"DemoOptions.ism")

   ELSE


DELETE


this.StandardPage("Deleted", ;

      "Invoice "+TRANSFORM(lnInvNo)+ ;

      " has been deleted!",,3,"DemoOptions.ism")

   ENDIF 

ENDIF 

ENDFUNC

The first thing that may strike you as odd is that this function calls both the Request.Form method and the Request.QueryString method to find the invoice number we’re looking for.  This is because exactly the same code is used for both the invoice form with the buttons and with the hyperlinks.  You must use Request.Form to retrieve the value of txtInvNo, which is submitted with the form, but when the code is called from a hyperlink, no form has been submitted.  All the information available must be sent to the function in the hyperlink itself.  Use the Request.QueryString method to retrieve values from the hyperlink.  You may recall that the end of the hyperlink had “?InvNo=2”.  The question mark indicates the start of parameters in the query string, and “InvNo” is the first parameter named.  When using named parameters, as we are here, you must pass the name of the parameter whose value you want to retrieve when you call the Request.QueryString method.
Looking a little further down in the code, you’ll see that once again, the code is checking if the owner of the invoice matches the user who is logged in.  If not, the “Hacker Alert” message is displayed.

Protecting Your Application from URL Hacking

What can you do to protect you application from this kind of hacking?  To start with, don’t use hyperlinks within your application unless you have to.  Remember, hyperlinks reveal more than you want the user to know about your application with what it displays on the status bar and the address bar.  If you do use hyperlinks, you should consider using some kind of encoding technique so that paths into your application aren’t quite as obvious.
The second thing you must do is to program defensively.  Don’t make any assumptions about the conditions under which a user got to any point in your code.

When you do use links, send an absolute minimum of information in the query string.  Any information about the user or session should be stored as session variables that are saved on the server, not sent in the query string.  This is an argument why the use of cookies can be more secure than not using them.  The use of cookies allows you to store information in session variables on the server instead of having to pass the information to the server on each hit, where it could be intercepted by a hacker.

If a user hacked the URL to run something they would have legitimate rights to run through normal means, you may not be able to detect the difference from within your code, but it shouldn’t matter in this case.

Does this mean that by using a form’s submit and POST operation that you’re safe from this kind of hacking?  Certainly not!  You must always be defensive and check input!  The next section drives home the reason why.

SQL Injection

SQL Injection is a term used to describe a technique for entering commands into places in your application that accept user input, but which create effects you didn’t intend.  This is best described with an example.

Create a file called Junk.txt in the application directory in which the demo for this session is running.  The contents of this file are irrelevant.  We just want the file to be around.  Open Windows Explorer and display your application directory.  You want to be able to see the Junk.txt file.
Run the demo and select the “SQL Injection” link.

When the screen comes up that asks for the owner name to display, enter this:

barbara] and trans(execscript("ERASE JUNK.TXT"))=[
Press the Submit button, and the listing of invoices will display.  But if you’re watching the Junk.txt file in Windows Explorer, you’ll see that it disappears.  Be aware, the junk.txt file that disappeared was in our application directory, not the virtual directory.  This is a directory that the Internet guest account has no rights to, and access is completely controlled through the application.  Yet, a user on the web was able to erase a file in this directory.
How did this happen?

Let’s take a look at the code that gets executed when you click the submit button from this screen.

FUNCTION SQLInjection

LOCAL lcOwner, lcOutput, lcWhere

lcOwner = UPPER(Request.Form("txtOwner"))

IF EMPTY(lcOwner)


lcWhere = []

ELSE


lcWhere = "where UPPER(OwnerName) = [" + lcOwner + "]"

ENDIF

SELECT * ;


FROM Invoices ;


&lcWhere ;


INTO CURSOR TmpInvoices

IF _tally > 0


lcOutput = this.StdHeader()


lcOutput = lcOutput + [<center>] + CRLF


lcOutput = lcOutput + [<br>Invoices for ] + lcOwner + [:] + CRLF


lcOutput = lcOutput + [<br><table>] + CRLF


lcOutput = lcOutput+[<th>Invoice</th><th>Date</th><th>Amount</th>]+CRLF


SCAN



lcOutput = lcOutput + [<tr>] + CRLF



lcOutput = lcOutput + [  <td>] + ;

TRANSFORM(TmpInvoices.InvNo) + [</td>] + CRLF



lcOutput = lcOutput + [  <td>] + DTOC(TmpInvoices.InvDate) + ;

[</td>] + CRLF



lcOutput = lcOutput + [  <td align="right">] + ;

STR(TmpInvoices.InvTotal,10,2) + [</td>] + CRLF



lcOutput = lcOutput + [</tr>] + CRLF


ENDSCAN


lcOutput = lcOutput + [</table>] + CRLF


lcOutput = lcOutput + [</center>] + CRLF


lcOutput = lcOutput+[<form method="POST" action="DemoOptions.ism">]+CRLF


lcOutput = lcOutput + [<p align="center">] + ;

<input type="submit" value="Done"></p>] + CRLF


lcOutput = lcOutput + [</form>] + CRLF


lcOutput = lcOutput + this.StdFooter()


Response.Write(lcOutput)

ELSE


this.StandardPage("No invoices found", ;

"There are no invoices for the owner entered",,5,"DemoOptions.ism")

ENDIF 

ENDFUNC

In the code above, pay close attention to how the user input is handled.  Notice that we just accept the user’s input unchecked and unfiltered, and incorporate it into the WHERE clause used in a SQL SELECT statement.  When this expression is macro-expanded, any command the user (or should I say “hacker”) has typed in gets executed.  In this example, the WHERE clause ended up like this:
"where UPPER(OwnerName) = [" +barbara] and trans(execscript("ERASE JUNK.TXT"))=[ + "]"
Now, just imagine if instead of “ERASE JUNK.TXT”, this command had contained “ERASE *.*” or “FORMAT C:”.  That should scare you!

Note that this is not a VFP problem, nor a problem with the web tool.  It’s a programming problem.  Your LAN apps are subject to this kind of abuse too, but you probably haven’t run into it because your application isn’t exposed to such a wide audience, and therefore isn’t susceptible to hackers.

There are a couple of things you can do to protect yourself from this kind of thing.

This wouldn’t happen if you didn’t include the “WHERE UPPER(OwnerName) =” part of the command in the variable.  In other words, if your SQL Statement looked like the example below, you wouldn’t have this problem.
SELECT * ;


FROM Invoices ;


WHERE UPPER(OwnerName) = lcOwner ;


INTO CURSOR TmpInvoices

But this is a simple case, and you may need to build more complex and dynamic WHERE clauses that require macro expansion.  Macro expansion in any form, or use of the EVALUATE function, could open you up for problems if you don’t check the contents.  Even with the original syntax, you could stay out of trouble by doing a couple of checks on the input.

First, in this application, the owner name is a 10 character field.  There is absolutely no reason the input should exceed that length.  So, the first thing we should do is check the length of what’s been entered.  If it’s greater than the maximum length of the field we’re matching it to, we must reject the input.

The second thing we should do is see if there are any characters in the input that are invalid for the field.  For example, there’s no reason our owner field should have characters like *&%^$#@, etc.

With these additions, the start of our function now looks like this:

FUNCTION SQLInjection

LOCAL lcOwner, lcOutput, lcWhere

lcOwner = UPPER(Request.Form("txtOwner"))

DO case

CASE LEN(lcOwner) > 10


this.ErrorMsg("Invalid entry","Invalid length for owner")


RETURN

CASE CHRTRAN(lcOwner,"<>][?~\!@#$%^&*()+=_'","") <> lcOwner


this.ErrorMsg("Invalid entry","Invalid characters in owner")


RETURN 

CASE CHRTRAN(lcOwner,["],[]) <> lcOwner


this.ErrorMsg("Invalid entry","Invalid characters in owner")

ENDCASE 

IF EMPTY(lcOwner)


lcWhere = []

ELSE


lcWhere = "where UPPER(OwnerName) = [" + lcOwner + "]"

ENDIF

The case statement that’s been added starts by making sure the input isn’t longer than 10 characters.  Depending on the sensitivity of your application, you may not even want to let the user know that the input is an invalid length, but simply reject it as “Invalid input”.
The second case in the statement checks for most of the invalid characters that could be entered.  We must delimit the string of invalid characters somehow, and here we’re using double quotes to delimit the string.  But the double quote character is also one we don’t want in the input, therefore, we have a third case that’s looking specifically for the double quotes in the string, using an alternate delimiter.

You probably don’t want to have to repeat code like this everywhere you need to check the input, so it’s best to create a centralized routine you can call.

For a much more involved discussion of SQL Injection and how to prevent it, read the article at http://www.spidynamics.com/papers/SQLInjectionWhitePaper.pdf.  
Microsoft has created a lockdown tool for IIS which you can read about at http://www.sans.org/rr/papers/65/301.pdf. You need to be aware of the issues discussed in this paper, whether you use the lockdown tool or not to secure your web server.
Conclusion
In this session, I’ve discussed some common mistakes when creating web applications as well as some easily missed precautions you should take for web applications.  I hope I’ve made you realize that developing applications for the web requires a very different mindset from the kind of application you’re probably used to writing.  I’ve barely scratched the surface here, when it comes to security issues, but you should now have an idea of the some of the kinds of issues you’ll be facing and how to deal with these.
� Thanks to Mike Cummings for making this observation.


� Available from � HYPERLINK "http://www.hentzenwerke.com" ��www.hentzenwerke.com�. 





